

## Bedienungsanleitung Modul M001 - Digital In/Out für KC85/2/3/4

### 1. Einleitung

Der Modul M001 DIGITAL IN/OUT (DIO) realisiert eine universell anwendbare digitale Ein/Ausgabe- und Zähler/Zeitgebereinheit für den KC 85/2 und seine Nachfolgetypen. Mit Hilfe des Moduls lassen sich anwenderspezifische Interfaceschaltungen oder Peripheriegeräte mit Parallelschnittstellen vom KC 85 steuern.

Desweiteren können durch den Modul extern anliegende Impulse gezählt bzw. zeitgesteuert in der extern angeschlossenen Peripherie Steuerereignisse ausgelöst werden. Damit können Probleme der Meß-, Steuer- und Regelungstechnik, sowie der Laborautomatisierung bearbeitet werden.

Das vorliegende Handbuch beschreibt den DIGITAL IN/OUT-Modul und seine Handhabung. Außerdem ist ein Applikationsbeispiel mit zugehöriger Software angegeben.

### 2. Modulhandhabung

#### 2.1 Stecken und Entfernen des Moduls

Der DIGITAL IN/OUT-Modul kann prinzipiell in jedem Modulsteckplatz betrieben werden, jedoch ist dabei die Modulpriorität in der gewählten Systemkonfiguration zu berücksichtigen. Es sind deshalb die nachfolgenden Informationen und die unter Pkt. 2.3. aufgeführten Hinweise zu beachten.

Die Modulprioritätskette muß geschlossen bleiben. Also sind zuerst im Grundgerät der Steckplatz 8 (rechts), danach der Steckplatz C (links) und anschließend weitere Steckplätze von Erweiterungsaufsätzen in aufsteigender Reihenfolge zu belegen.

#### **Achtung !**

Das Stecken und Entfernen des Moduls darf nur im ausgeschalteten Zustand des Systems vorgenommen werden!

Damit ergeben sich folgende Handgriffe für das Stecken des Moduls:

- a) Den Computer ausschalten.
- b) Die Kappe des Modulschachtes abnehmen. Hierzu muß die Kappe an den gegenüberliegenden Griffflächen leicht zusammen gedrückt werden.
- c) Den Modul bis zum fühlbaren Einrasten einschieben. Der Modul ist dann richtig kontaktiert, wenn sein hervorstehender Rand unmittelbar an der Gerätewand anliegt.
- d) Nun kann der Computer eingeschaltet werden.

Bevor die weitere Inbetriebnahme beschrieben wird, soll an dieser Stelle gleich das Entfernen des Moduls aus dem System erläutert werden.

Zum Entfernen des Moduls sind folgende Schritte notwendig:

- a) Den Computer ausschalten.
- b) Den linken und den rechten Zeigefinger unter den Modulkopf legen und mit den Daumen die seitlich am Modul befindlichen Hebel gleichzeitig nach unten drücken. Dabei rastet der Modul aus und wird etwa einen Zentimeter aus dem Gerät herausgeschoben. Nun kann der Modul leicht aus dem Schacht gezogen werden.
- c) Die Kappe auf die Schachöffnung stecken.

#### 2.2. Modulzuweisung und Steuerbyte

Der DIGITAL IN/OUT-Modul hat im KC-System zwei Betriebszustände:

- INAKTIV Diode leuchtet nicht. Der Modul ist vom Prozessor getrennt.
- AKTIV Diode leuchtet. Falls es die Modulpriorität erlaubt (siehe Pkt. 2.3.), können über den aktivierten Modul Daten gesendet und/oder empfangen werden.

Der gewünschte Betriebszustand wird über das Kommando

SWITCH mm kk (Parameter durch Leerzeichen voneinander getrennt)  
eingestellt. Die beiden Parameter dieses Kommandos realisieren:

- mm Mitteilung an das System, in welchem Modulschacht der zuzuweisende Modul kontaktiert ist. Dabei ist die erste Stelle von mm die Nummer des Aufsatzes (im Grundgerät ist diese Stelle Null und kann weggelassen werden). Die zweite Stelle von mm ist die Steckplatzadresse. Im Grundgerät existieren nur die Steckplatzadressen 8 (rechter Schacht) und C (linker Schacht).
- kk Schalten des Moduls. Die erste Stelle von kk ist Null und kann weggelassen werden. Die zweite Stelle von kk kann sein:
  - 0 - Modul inaktiv
  - 1 - Modul aktiv.

Beispiel: Der DIGITAL IN/OUT-Modul steckt im Schacht 8 und soll aktiviert werden.  
Es ist einzugeben:

SWITCH 8 1

Nach Betätigen der ENTER-Taste muß die Diode des M001 DIGITAL IN/OUT leuchten.  
Der Modul ist in den aktiven Zustand geschaltet. Auf dem Bildschirm erscheinen folgende Informationen:

08 EF 01

Mit dem ersten Byte wird angezeigt, daß der Modul im Steckplatz 8 angesprochen wurde. Das zweite Byte ist das Strukturbyte des Moduls, der im angesprochenen Modulschacht steckt. Durch das dritte Byte bestätigt der Computer die Eingabe des Parameters kk.

Jeder Modul erhält ein für ihn charakteristisches Strukturbyte. Dieses Strukturbyte widerspiegelt den Modultyp bzw. die innere Strukturierung des Moduls. Das Strukturbyte kann durch den Prozessor (auch im inaktiven Zustand des Moduls) gelesen werden. Dadurch kann sich der Nutzer jederzeit in einem ausgebauten System einen Überblick über die verfügbaren Module verschaffen und in Abhängigkeit davon seine Entscheidungen treffen.  
Der DIGITAL IN/OUT-Modul besitzt, wie das Beispiel zeigt, das Strukturbyte:

EFH.

Das Herausschalten des Moduls aus dem System (Betriebszustand INAKTIV) erfolgt über:

SWITCH mm 00 (für mm ist die Modulsteckplatzadresse einzugeben)

Auf dem Bildschirm erscheint:

mm EF 00

Hinweis:

Nach Betätigen der RESET-Taste bleibt das Steuerbyte des Moduls erhalten. Ein aktiv geschalteter DIO-Modul ist weiterhin im aktiven Zustand. Lediglich die beiden Schaltkreise PIO und CTC (siehe Pkt. 3.) werden durch das Signal /RESET in den Ausgangszustand versetzt und müssen programmtechnisch neu initialisiert werden.

### 2.3.Modulprioritätssteuerung

Jeder DIGITAL IN/OUT-Modul ist mit einer Prioritätssteuerung ausgestattet, die seine Eingliederung in das KC-System ermöglicht. Dadurch können mehrere DIGITAL IN/OUT-Module in das KC-System eingeordnet werden. Ist dies der Fall, gilt folgende Rangfolge:

Derjenige aktivierte Modul, der sich auf dem Modulsteckplatz mit der niedrigsten Steckplatznummer befindet, besitzt gegenüber den anderen aktiv geschalteten DIGITAL IN/OUT-Modulen die höchste Priorität. Ein im Steckplatz 8 des Grundgerätes kontaktierter DIO-Modul ist danach höher priorisiert als ein DIO-Modul im Steckplatz C. Sind in beiden Schächten DIO-Module aktiviert, so kann der Modul im Schacht C nur dann benutzt werden, wenn der Modul im Schacht 8 INAKTIV geschaltet wird.

Verallgemeinert gilt:

Durch INAKTIV-Schalten des in der Rangfolge höherpriorisierten DIO-Moduls, kann der AKTIV geschaltete DIO-Modul mit der nächstniederen Priorität benutzt werden. Das AKTIV- bzw. INAKTIV-Schalten erfolgt über das SWITCH-Kommando (siehe Pkt. 2.2.) bzw. über das entsprechende Betriebssystem-Unterprogramm (vgl. Systemhandbuch KC85).

### 3. Technische Beschreibung

#### 3.1. Realisierung der Ein/Ausgabe-Einheit

##### 3.1.1. Aufbau des DIO-Moduls

Neben der integrierten Modulsteuerung besteht der DIO-Modul aus den beiden Hauptkomponenten PIO (Parallel Ein/Ausgabe Schaltkreis) und CTC (Zeitgeber Schaltkreis). Über den Peripherieanschluß stehen die beiden 8-Bit-PIO-Ports (A0-A7 und B0-B7) und die den Toren zugeordneten Quittungssignale (ASTB, ARDY, BSTB und BRDY) zur Verfügung. Desweiteren kann der Anwender über 2 CTC-Kanäle an der Peripherie verfügen, die zur Signalzählung oder Steuerung der extern angeschlossenen Interfaceschaltung benutzbar sind. Herausgeführt sind die CTC-Kanäle 0 und 1 (CLK/TRG0, CLK/TRG1, ZC/TO0 und ZC/TO1). Die CTC-Kanäle 2 und 3 sind intern kaskadiert, wobei ZC/TO2 mit CLK/TRG3 verbunden ist. Diese beiden Kanäle können damit als Softwareuhr programmiert werden. Dabei ist der CTC-Kanal 2 als Zeitgeber und der CTC-Kanal 3 als interruptgesteuerter Zähler zu programmieren. Die CTC-Kanäle 2 und 3 können aber auch unabhängig voneinander in der Betriebsart Zeitgeber verwendet werden.

##### 3.1.2. Kurzbeschreibung der Schaltkreise PIO und CTC

Den Hauptbestandteil des DIO-Moduls bilden:

- der Parallel Ein/Ausgabe-Schaltkreis PIO (Parallel Input/Output Controller U855D) und
- der Zeitgeberschaltkreis CTC (Counter Timer Circuit U857D)

Der PIO ist ein hochintegrierter Schaltkreis des U880-Prozessorsystems, der vorwiegend für den Einsatz in der Steuerungs- und Regelungstechnik vorgesehen ist. Der Schaltkreis ist weitgehend frei programmierbar und kann den peripheren Bedingungen angepaßt werden. Der PIO ermöglicht die parallele digitale Ein- und Ausgabe von Daten zwischen der CPU und der angeschlossenen Peripherie. Er enthält zwei TTL-kompatible 8-Bit-Tore (Ports), deren Betriebsart durch entsprechende Steuerbefehle programmtechnisch festgelegt werden kann. Der Schaltkreis verfügt über eine vollständige Logik zur Interruptkaskadierung und -priorisierung.

Weitere wichtige Merkmale des Schaltkreises sind:

- Interruptmöglichkeit im Quittungsbetrieb (Handshaking) für schnelle Anforderungsbearbeitung. Die beiden 8-Bit-bidirektionalen Ports sind mit Einrichtungen für Quittungsbetrieb versehen.
- Programmierbare Betriebsarten
  - . Byte-Ausgabe (Mode 0)
  - . Byte-Eingabe (Mode 1)
  - . Byte-Ein/Ausgabe (bidirektionaler Betrieb, nur für Port A möglich, Mode 2)
  - . Bit-Ein/Ausgabe (Mode 3)
- Automatische Interruptvektorerzeugung und Prioritätskodierung durch Kaskadierung
- Alle Ein- und Ausgänge sind TTL-kompatibel

Der CTC-Schaltkreis gehört ebenfalls zur U880-Systemschaltkreisfamilie. Der CTC ist ein Vier-Kanal-Zähler/Zeitgeberbaustein, dessen Betriebsart und Operationsparameter programmtechnisch festgelegt werden und in bestimmten Grenzen variabel sind.

Der Schaltkreis ist durch folgende charakteristische Merkmale gekennzeichnet:

- Vier voneinander unabhängig programmierbare 8-Bit-Zähler/16-Bit-Zeitgeberkanäle.
- Jeder Kanal ist wahlweise als Zähler oder Zeitgeber programmierbar.
- Aktuelle Werte (Zähler- bzw. Zeitgeberzustände) sind über den Datenbus abrufbar.
- In der Betriebsart Zeitgeber verwendet jeder Kanal einen Vorteiler, mit dem er die durch 16 oder 256 geteilten Systemtakte als Zeitbasis benutzt.
- Ein als Zeitgeber programmierter Kanal kann wahlweise durch positive oder negative Triggerimpulse gestartet werden.
- In der Betriebsart Zähler ist die aktive Zählflanke wählbar.
- Die 4 Kanäle bilden 4 unabhängige Einheiten.
- Nulldurchgänge der Kanäle können Interrupts anfordern, d.h. periphere Zustände können Interrupts auslösen bzw. zur Steuerung der Peripherie benutzt werden.
- Die 4 Kanäle sind intern zu einer Interruptprioritätskette geschaltet, wobei die Prioritätsreihenfolge festliegt.
- Die maximale Zählfrequenz in der Betriebsart Zähler beträgt  $f_c/2$ . ( $f_c$ =Systemtaktfrequenz des Computers; 1,751938 MHz)
- Die Kanäle sind zur Erweiterung des Zählumfanges und gegenseitigen Triggerung kaskadierbar.
- Alle Ein- und Ausgänge des CTC sind TTL-kompatibel.

Detaillierte Informationen zur Funktionsweise und Programmierung der Schaltkreise sind aus den unter Pkt. 7. angegebenen Literaturhinweisen zu entnehmen.

### 3.2. Interruptprioritätskette

Der DIO-Modul kann voll interruptgesteuert im KC-System betrieben werden. Dabei ist der CTC höherpriorisiert als der PIO.

### 3.3. Interne Adressen

Für den DIO-Modul sind im KC-System die in Tafel 1 angegebenen Adressen reserviert.

Ein/Aus- gabekanal	CTC- Kanal	PIO- Kanal	Bedeutung
00H	0		Steuerwort/Zeitkonstante
01H	1		Steuerwort/Zeitkonstante
02H	2		Steuerwort/Zeitkonstante
03H	3		Steuerwort/Zeitkonstante
04H		A	Datenwort
05H		B	Datenwort

06H	A	Steuerwort
07H	B	Steuerwort

Tafel 1: Interne Adressen des DIO-Moduls

### 3.4. Peripherieanschlußbedingungen

#### 3.4.1. Peripherieanschlußbelegung

Der Modul verfügt über eine 26-polige Buchsenleiste für direkten Anschluß. Die Buchsenleiste ist das Koppellement zur extern angeschlossenen Peripherie. Im Bild 2 ist die Peripherieanschlußbelegung dargestellt. Die Darstellung zeigt den Blick auf die Frontplatte des Moduls.

									ZC/	ZC/													
Port A	_	0	1	2	3	4	5	6	7	STB	RDY	TO0	TO1	A									
		1	2	3	4	5	6	7	8	9	10	11	12	13									
Port B	_	0	1	2	3	4	5	6	7	STB	RDY	CLK/CLK/	B										
										TRG0	TRG1												

Bild 2: Peripherieanschlußbelegung

- Pin 1 A/B - Masse
- Pin 2 bis 9 A - PIO Port A (A0-A7)
- Pin 2 bis 9 B - PIO Port B (B0-B7)
- Pin 10 A - STB A (PIO Port A Quittungssignal, Eingang)
- Pin 10 B - STB B (PIO Port B Quittungssignal, Eingang)
- Pin 11 A - RDY A (PIO Port A Quittungssignal, Ausgang)
- Pin 11 B - RDY B (PIO Port B Quittungssignal, Ausgang)
- Pin 12 A - ZC/TO0 (CTC Kanal 0, Nulldurchgangssignal, Ausgang)
- Pin 13 A - ZC/TO1 (CTC Kanal 1, Nulldurchgangssignal, Ausgang)
- Pin 12 B - CLK/TRG0 (CTC Kanal 0, Clock/Trigger-Eingang)
- Pin 13 B - CLK/TRG1 (CTC Kanal 1, Clock/Trigger-Eingang)

#### 3.4.2. Elektrische Bedingungen

Für die am Peripherieanschluß herausgeführten Signale (PIO Port A und B mit Quittungssignalen und die CTC Kanäle 0 und 1) gelten folgende elektrischen Bedingungen:

- TTL-Pegel

(UOH > 2,4 V , UOL < 0,4 V

UIH > 2,0 V , UIL < 0,8 V)

- 1 TTL-Last

(IOH < -0,25 mA

IOL < 1,6 mA)

#### 3.4.3 Eingangsschutz

Alle Peripherieanschlußsignale sind mit einer Schutzbeschaltung versehen. Diese Schaltung gewährleistet einen Eingangsschutz der Schaltkreise PIO und CTC in einem Spannungsbereich von

-3 V bis +8 V.

#### 3.4.4. Leitungslänge

Ohne zusätzliches Treiberinterface für die PIO- bzw. CTC-Signale, also bei direkter Kopplung von CTC und PIO mit der externen Peripherie, sollte unter Einhaltung der unter Pkt 3.4.2. genannten Pegel und Lastbedingungen eine Leitungslänge von 1 m nicht überschritten werden.

#### 4. Zubehör

Als Zubehör erhalten Sie einen Bauteilsatz Stecker. Dieser besteht aus folgenden Einzelteilen:

- 2 x Griffschale 11 TGL 28 597
- 2 x Bügel 2 TGL 28 597
- 2 x Zylinderschraube BM 1,6x10
- 2 x Sechskantmutter M 1,6
- 1 x Leiterplatte Schl.-Nr. 48 258
- 1 x Kodierschieber 1 TGL 29 331/01

#### 5. Anschlußempfehlung für Peripheriestecker

Als Peripherieanschluß besitzt der Modul eine 26-polige Buchsenleiste für direkten Anschluß. Durch den als Zubehör (siehe Pkt. 4.) mitgelieferten Peripheriestecker kann die Verbindung zwischen externer Peripherie und dem Modul realisiert werden.

Der beiliegende Kodierschieber gewährleistet die Verdrehsicherheit des Peripheriesteckers. Die Leiterplatte des Steckers ist mit einem Kodierschlitz versehen. Der Kodierschieber ist dazu an der entsprechenden Stelle der Buchsenleiste aufzustecken.

Um eine Vereinheitlichung der Kodierung der Koppellemente zu erreichen, wird Ihnen empfohlen den Kodierschieber zwischen die Anschlüsse 5 und 6 der Buchsenleiste einzurasten.

Die 2 Bügel dienen zur Verriegelung der Koppellemente, um ungewolltes Herausziehen zu verhindern. Dazu werden die Bügel am Kragen der Buchsenleiste eingehängt, der Peripheriestecker kontaktiert und die Bügel über die Haltenasen der Griffelemente des Steckers geschoben.

#### 6. Anwendungsbeispiel

##### **CENTRONICS - Schnittstelle**

Die CENTRONICS-Schnittstelle ist eine parallele Schnittstelle für TTL-Signale, die einen bitparallelen, byteseriellen Informationsaustausch ermöglicht.

Die Schnittstelle ist vornehmlich für den Anschluß von Druckern vorgesehen. Es können damit aber auch andere Peripheriegeräte, wie Plotter, intelligente Meßgeräte oder EPROM-Programmiergeräte usw., betrieben werden. Im folgenden Beispiel soll die Ausgabe von ASCII-Zeichen an einen Drucker, der über ein CENTRONICS-Interface verfügt, realisiert werden. Dabei gelten folgende Übertragungsbedingungen:

- Übergabe eines ASCII-Zeichens parallel auf den Datenleitungen DATA1 - DATA8.
- Ausgabe eines Low-Impulses auf der Leitung /STROBE.
- Der Drucker schaltet die Leitung BUSY auf High-Pegel und zeigt dem Computer den Datenempfang an.
- Mit Low-Pegel auf der Leitung BUSY zeigt danach der Drucker dem Computer die Empfangsbereitschaft für ein neues ASCII-Zeichen an.

Die Schnittstellenbelegung für das CENTRONICS-Interface am M001 zeigt Tafel 2.

Anschluß    Bezeichnung    Leitung

1A,1B	Masse	Masse
2A	PIO A0	DATA1
3A	PIO A1	DATA2
4A	PIO A2	DATA3
5A	PIO A3	DATA4
6A	PIO A4	DATA5
7A	PIO A5	DATA6
8A	PIO A6	DATA7
9A	PIO A7	DATA8
2B	PIO B0	/STROBE
4B	PIO B2	BUSY

Tafel 2: Schnittstellenbelegung

Die Anschlußleitungen sollten eine Länge von 1m nicht überschreiten. Außerdem ist es empfehlenswert, die Datenleitungen mit jeweils einer Masseleitung zu verdrillen.

Zur softwareseitigen Einbindung ist auf den folgenden Seiten das Assemblerlisting des CENTRONICS-Treibers dargestellt.

Das Programm gestattet sowohl den zeichenweisen Ausdruck über den Anwenderkanal 2, dies ist im BASIC über PRINT#2 bzw. LIST#2, als auch ein Mitprotokollieren aller Bildschirmausgaben auf den Drucker. Das Mitprotokollieren wird über Tastenkombination 'SHIFT CLEAR' ein- bzw. ausgeschaltet. Bei der Initialisierung des Programmes kann der Modulschacht angegeben werden, in welchem der Modul M001 steckt. Weiterhin kann ausgewählt werden, über welchen Anwenderkanal ausgegeben werden soll. Dies ist z.B. sinnvoll, wenn gleichzeitig im System auch eine V.24-Schnittstelle eingesetzt wird. Der Aufruf des Programmes erfolgt aus der Menüeingabe des Betriebssystems.

Folgendes Beispiel verdeutlicht dies:

```
%CENTRON 8 2
```

Hier wird der Modul im Schacht mit der Adresse 8 für den Anwenderkanal 2 (im BASIC #2) initialisiert. Dies entspricht der Initialisierung im Assemblerlisting auf den folgenden Seiten.

Steht ein DEVELOPMENT-Modul M027 zur Verfügung, so kann das Listing eingegeben und übersetzt werden. Anderenfalls kann auch eine Eingabe des Maschinencodes mittels des CAOS-Kommandos 'MODIFY' ab Adresse BA00

erfolgen. In beiden Fällen wird das Maschinenprogramm als selbststartendes Programm auf Kassette abgespeichert.

Die Kommandoeingabe dafür lautet:  
%SAVE BA00 BB00 BA02  
NAME: CENTRONICOM

Alle weiteren Informationen zum Programm können dem Listing entnommen werden.

```
3000      ,*****
3000      ;
3000      ORG 0BA00H
BA00     ; CENTRONIC- TREIBER
BA00     ; 4.9.86
BA00     ; VEB MIKROELEKTRONIK
BA00     ; MUEHLHAUSEN
BA00     ,*****
BA00     ;
BA00     ;VEREINBARUNGEN:
BA00     CAOS EQU 0F003H
BA00     MODU EQU 26H ;UP-NR. MODUL
BA00     CRT EQU 0 ;UP-NR. CRT
BA00     ;SYSTEMZELLEN CAOS
BA00     ARGN EQU 0B781H
BA00     STBT EQU 0B7A2H ;STEUERBYTE
BA00     OUTAB EQU 0B7B9H ;OUT-CHANAL
BA00     UOUT1 EQU 0B7BEH ;USER OUT 1
BA00     UOUT2 EQU 0B7C4H ; " 2
BA00     HCADR EQU 0B799H ;SHIFT CLEAR
BA00     CRR EQU 0DH
BA00     ;
BA00     MKENN EQU 0EFH ;MODULKENNYBYTE
BA00     PIOA EQU 04H ;PIO KANAL A DAT
BA00     PIOAC EQU 06H ; STEUERWORT
BA00     PIOB EQU 05H ;PIO B DATEN
BA00     PIOBC EQU 07H ; STEUERWORT
BA00     ;
BA00     ,*****
BA00     ;
BA00     ;PARAMETERUEBERGABE BEI AUFRUF:
BA00     ;CENTRON MODULSCHACHT USER-OUT
BA00     ; 8 u.s.w. 2/3
BA00     ,*****
BA00     ;
BA00 00   HCPZ: DEFB 0 ;USER-KANAL
BA01     ;
BA01     ; MODULSCHACHT 8
BA01 08   MODSCH: DEFB 8 ;MODULSCHACHT
BA02     ;-----
BA02 C324BA JP START ;SELBSTSTARTADR.
BA05     ;-----
BA05 7F7F DEFW 7F7FH ;PROLOG
BA07 43454E54 DEFM 'CENTRON'
BA0E 01   DEFB 1
BA0F 3A81B7 LD A,(ARGN)
BA12 FE02 CP 2
BA14 380E JR C,START
BA16 7D   LD A,L ;ARG1,LOW
BA17 3201BA LD (MODSCH),A ;SCHACHT
BA1A 7B   LD A,E ;USEROUT
BA1B D602 SUB 2
BA1D 3811 JR C,ERROR
BA1F E601 AND 1
BA21 3200BA LD (HCPZ),A
BA24     ; SELBSTSTARTADRESSE
```



```

BA24      ;+++++
BA24 3A01BA START: LD  A,(MODSCH)
BA27 47      LD  B,A
BA28 0E80      LD  C,80H
BA2A ED78      IN  A,(C)
BA2C FEEF      CP  MKENN
BA2E 2805      JR  Z,ST1
BA30 CD03F0 ERROR: CALL CAOS ;FEHLERABBRUCH
BA33 19      DEFB 19H
BA34 C9      RET
BA35      ;
BA35 68      ST1: LD  L,B
BA36 3E02      LD  A,2
BA38 1601      LD  D,1
BA3A 5A      LD  E,D
BA3B CD03F0      CALL CAOS ;MODULSWITCH
BA3E 26      DEFB MODU
BA3F 3EFF      LD  A,0FFH
BA41 D306      OUT  PIOAC,A ;BIT E/A
BA43 3E00      LD  A,0
BA45 D306      OUT  PIOAC,A ;ALLE BIT OUT
BA47 3EFF      LD  A,0FFH
BA49 D307      OUT  PIOBC,A ;BIT E/A
BA4B 3E04      LD  A,4
BA4D D307      OUT  PIOBC,A ;BIT0 OUT
BA4F 3E01      LD  A,1
BA51 D305      OUT  PIOB,A ;STROBE PASSIV
BA53 215FBA      LD  HL,HCOP
BA56 2299B7      LD  (HCADR),HL
BA59 E5      PUSH HL
BA5A D5      PUSH DE
BA5B 1830      JR  HACO1
BA5D      ;
BA5D 02      ZWEI: DEFB 2
BA5E 03      DREI DEFB 3
BA5F      ;
BA5F      ;PROTOKOLLFUNKTION UEBER SHIFT CLEAR
BA5F      ;ODER BILDSCHIRMAUSDRUCK
BA5F E5      HCOP: PUSH HL
BA60 D5      PUSH DE
BA61 2AB9B7      LD  HL,(OUTAB) ;KANAL
BA64 34      INC  (HL)
BA65 35      DEC  (HL)
BA66 2025      JR  NZ,HACO1 ;CRT ->
BA68 1195BA      LD  DE,PROT ;CRT+KAN
BA6B 215DBA      LD  HL,ZWEI
BA6E 3A00BA      LD  A,(HCPZ)
BA71 A7      AND  A
BA72 2803      JR  Z,HACO3 ;USER2->
BA74 215EBA      LD  HL,DREI
BA77 3A00BA HACO3: LD  A,(HCPZ)
BA7A A7      AND  A
BA7B 2806      JR  Z,HACO2
BA7D ED53C4B7      LD  (UOUT2),DE ;EINTR.3
BA81 1804      JR  HACO4
BA83      ;
BA83 ED53BEB7 HACO2: LD  (UOUT1),DE ;EINTR.2
BA87 22B9B7 HACO4: LD  (OUTAB),HL
BA8A D1      HAPOP: POP DE
BA8B E1      POP HL
BA8C C9      RET
BA8D      ;
BA8D 2199BA HACO1: LD  HL,ZCRT ;CRT

```

```

BA90 119BBA      LD  DE,SDD
BA93 18E2        JR  HACO3
BA95      ;PROTOKOLLAUSGABE
BA95 F5  PROT:  PUSH  AF
BA96 CD03F0     CALL CAOS  ;BILDSCHIRMAUSG.
BA99 00  ZCRT:  DEFB  00H
BA9A F1        POP  AF
BA9B      ;*****
BA9B      ; ZEICHENAUSGABE ZEICHEN IN A
BA9B FE09     SDD:  CP   09H  ;RIGHT?
BA9D 280D     JR   Z,SD4
BA9F F5      PUSH  AF
BAA0 3AA2B7   LD   A,(STBT)
BAA3 CB5F     BIT   3,A  ;INTERPRET?
BAA5 280F     JR   Z,SD9  ;YES
BAA7 F1      POP  AF
BAA8 FE7F     CP   7FH  ;AUSBLENDEN 7FH
BAAA 2002     JR   NZ,SD3
BAAC 3E20     SD4: LD   A,20H
BAAE FE20     SD3: CP   20H  ;SPACE
BAB0 301A     JR   NC,SD1
BAB2 3E5F     LD   A,5FH
BAB4 1816     JR   SD1
BAB6 F1      SD9: POP  AF
BAB7 FE0D     CP   0DH
BAB9 2011     JR   NZ,SD1
BABB      ;ZEILENPOSIT 8* SPACE
BABB C5      SD0: PUSH  BC
BABC F5      PUSH  AF
BABD CDCCBA   CALL  SD1
BAC0 0608     LD   B,8
BAC2 3E20     SDLOP: LD  A,20H
BAC4 CDCCBA   CALL  SD1
BAC7 10F9     DJNZ SDLOP
BAC9 F1      POP  AF
BACA C1      POP  BC
BACB C9      RET
BACC      ;DIREKTE BYTEAUSGABE
BACC C5      SD1: PUSH  BC  ;AUSGABE DATEN
BACD F5      PUSH  AF
BACE DB05     SDA: IN  A,PIOB
BAD0 CB57     BIT   2,A
BAD2 2808     JR   Z,SDB  ;BUSY?
BAD4 3E05     LD   A,5
BAD6 CD03F0   CALL  CAOS
BAD9 14      DEFB  14H  ;WARTEN
BADA 18F2     JR   SDA
BADC      ;
BADC F1      SDB: POP  AF
BADD D304     OUT  PIOA,A ;DATEN
BADF 3E00     LD   A,0
BAE1 D305     OUT  PIOB,A ;STROBE AKTIV
BAE3 3E01     LD   A,1
BAE5 D305     OUT  PIOB,A ;STROBE PASSIV
BAE7 C1      POP  BC
BAE8 C9      RET
BAE9      ;
BAE9      ;  E N D E

```

Um sich vertiefend mit der Anwendung und Programmierung der U880-Systemschaltkreise zu befassen, sei auf folgende Literatur hingewiesen:

- /1/ "Mikroprozessorsystem der II. Leistungsklasse"  
Hefte CPU, CTC, PIO  
VEB Mikroelektronik "Karl-Marx" Erfurt
- /2/ Kieser, H.; Meder, M. "Mikroprozessortechnik"  
VEB Verlag Technik Berlin
- /3/ Claßen, L. "Programmierung des Mikroprozessorsystems  
U 880-K 1520"  
Reihe Automatisierungstechnik  
VEB Verlag Technik Berlin
- /4/ Berthold, H. "Mikroprozessoren-Mikroelektronische  
Schaltkreise und ihre Anwendung" Teil 1-3  
electronica Bd. 186-188, Militärverlag
- /5/ Lampe, B.; Jorke, G.; Wengel, N. "Algorithmen der Mikro  
rechentechnik"  
VEB Verlag Technik Berlin